


課題 立体感のある球

円の中心 (x, y) と半径 r を少しずつ変化させることにより立体感のある球を描画する。 r_{max} は半径の最大値。



● プログラム (KA641. bas)

```

1  ' << KA641. bas >>
2  ' 立体感のある球
3  '
4  ' グラフィック画面の背景色を白に設定。
5  BackColor="White"
6  '
7  ' グラフィック画面のサイズ（縦200ピクセル、横200ピクセル）を設定。
8  GScreen(300,300)
9  '
10 ' 座標の設定。画面左下隅の座標を(-1.2,-1)、右上隅の座標を(1,1.2)
11 ' とする。
12 Window(-1.2,-1)-(1,1.2)
13 MathGraph On
14 '
15 ' 初期設定。
16 ' 中心(X,Y),最大半径RMAX,半径Rの設定。
17 X=0: Y=0: RMAX=0.8: R=RMAX
18 ' 中心x座標の移動量DX, 中心y座標の移動量DY、半径Rの変化量DR。
19 DX=0.02: DY=0.04: DR=0.04
20 '
21 ' 球の描画。
22 While R > 0
23   If Sqr(X*X+Y*Y)+R < RMAX Then
24     ForeColor="Black"
25     Circle(X,Y),R: '(X,Y)を中心とする半径Rの円を黒色で描く。
26   EndIf
27   '
28   ' 移動量、変化量の更新。
29   DX=DX*0.8


```

```

30  DY=DY*0.9
31  DR=DR*1.1
32  ' 中心、半径の更新。
33  X=X+DX
34  Y=Y+DY
35  R=R-DR
36  Wend
37  End

```

実行結果



グラフィック画面の点(x, y)の色を指定するには、RGB値（#RRGGBBの形）を使う方法がある。RR, GG, BBは、それぞれ16進数で00～FFの値をもつ。

#RRGGBBは、#000000から#FFFFFFの値を取る。

最初の2桁16進数はRed(赤)の輝度，2番目の2桁16進数はGreen(緑)の輝度，最後の2桁16進数はBlue(青)の輝度を表す。

色	RGB値
Black	#000000
Navy	#000080
Green	#008000
Teal	#008080
Maroon	#800000
Purple	#800080
Olive	#808000
Silver	#C0C0C0
Gray	#808080
Blue	#0000FF
Lime	#00FF00
Cyan	#00FFFF
Red	#FF0000
Fuchsia	#FF00FF
Yellow	#FFFF00
White	#FFFFFF

●プログラム (KA642. bas)

```

1  ' << KA642. bas >>
2  ' 立体感のある球
3  '
4  ' グラフィック画面の背景色を白に設定。
5  BackColor="White"
6  '
7  ' グラフィック画面のサイズ（縦200ピクセル、横200ピクセル）を設定。
8  GScreen(300,300)
9  '
10 ' 座標の設定。画面左下隅の座標を(-1,-1)、右上隅の座標を(1,1)とする。
11 Window(-1.2,-1)-(1,1.2)
12 MathGraph On
13 '
14 ' 初期設定。
15 ' 中心(X,Y),最大半径RMAX,半径Rの設定。
16 X=0: Y=0: RMAX=0.8: R=RMAX
17 ' 中心x座標の移動量DX,中心y座標の移動量DY、半径Rの変化量DR。
18 DX=0.02: DY=0.04: DR=0.04
19 '
20 ' 色の初期値入力。
21 Input"赤の輝度初期値(0~255[0])";RED$
22 If RED$ = "" Then RED$="0": RED=0 Else RED=Val(RED$)
23 Input"赤の輝度増分(0~255[0])";RI$
24 If RI$ = "" Then RI$="0": RI=0 Else RI=Val(RI$)
25 Input"緑の輝度初期値(0~255[0])";GREEN$
26 If GREEN$ = "" Then GREEN$="0": GREEN=0 Else GREEN=Val(GREEN$)
27 Input"緑の輝度増分(0~255[0])";GI$
28 If GI$ = "" Then GI$="0": GI=0 Else GI=Val(GI$)
29 Input"青の輝度初期値(0~255[0])";BLUE$
30 If BLUE$ = "" Then BLUE$="0": BLUE=0 Else BLUE=Val(BLUE$)
31 Input"青の輝度増分(0~255[0])";BI$
32 If BI$ = "" Then BI$="0": BI=0 Else BI=Val(BI$)
33 '
34 ' パラメータの表示。
35 GLocate(-1.2,1.15): W$="立体感のある球": Gprint W$
36 GLocate(-1.2,1.05): W$="赤の輝度 "+RED$+" 増分 "+RI$: Gprint W$
37 GLocate(-1.2,0.95): W$="緑の輝度 "+GREEN$+" 増分 "+GI$: Gprint W$
38 GLocate(-1.2,0.85): W$="青の輝度 "+BLUE$+" 増分 "+BI$: Gprint W$
39 '
40 '
41 ' 球の描画。
42 While R > 0
43   If Sqr(X*X+Y*Y)+R < RMAX Then
44     W=(RED*2^16+GREEN*2^8+BLUE)+2^24
45     W$=Right$(Hex$(W),6): ' 関数Hex$(W)でWの値を16進数の文字列に
46                          ' 変換する。
47     ForeColor="#" + W$
48     Circle(X,Y),R: ' (X,Y)を中心とする半径Rの円を前景色で描く。

```

```

49   Paint(X,Y):      ' (X,Y)を含む境界色で囲まれた領域を前景色で塗りつぶす。
50   EndIf
51   '
52   ' 移動量、変化量の更新。
53   DX=DX*0.8
54   DY=DY*0.9
55   DR=DR*1.1
56   ' 色の輝度を増加する。
57   RED=(RED+RI) Mod 256
58   GREEN=(GREEN+GI) Mod 256
59   BLUE=(BLUE+BI) Mod 256
60   ' 中心、半径の更新。
61   X=X+DX
62   Y=Y+DY
63   R=R-DR
64 Wend
65 End

```

